Equations (cut/paste from eng-tips)
1: 0 = F2*cos(phi1 + theta6) - F1*cos(theta6) + To*cos(lamda4) - To*cos(lamda5)
2 : 0 = F2*sin(phi1 + theta6) - F1*sin(theta6) + To*sin(lamda4) + To*sin(lamda5)
3 : 0 = - F2*cos(phi1 + theta6) - F3*cos(phi2 + theta7) - Fs*cos(psiL)
4 : 0 = - F2*sin(phi1 + theta6) - F3*sin(phi2 + theta7) - Fs*sin(psiL)
5 : 0 = F3*cos(phi2 + theta7) - F4*cos(theta7) + To*cos((3*pi)/2 + lamda2) + To*cos(pi/2 - lamda3)
6 : 0 = To*sin(pi/2 - lamda3) + F3*sin(phi2 + theta7) - F4*sin(theta7) + To*sin((3*pi)/2 + lamda2)

Form of equation
AX = B

Alternate form:
AX - B = 0
Call each row of the LHS "expre" for expression

CONTENTS

1 - Initialization
2 - Convert equations to augmented matrices A(6x5) and b (6x1)
3 - Demonstrate syntax for deleting a row (deleting an equation)
4 - Generate solution matrices by deleting one row at a time
5 - Compare results using ARBITRARY values
Conclusion: the results do not agree when deleting one row (equation) at a time

1 - INITIALIZATION

```
reset()
with(linalg)
```

```
with(linalg)
```

2 = CONVERT EQUATIONS TO AUGMENTED MATRICES A(6x5) and b(6x1)

```
expre:=matrix(6,1): // Initializes expre... expressions for eqns 1, 2 etc
```

```
expre(1,1):=F2*cos(phi1 + theta6) - F1*cos(theta6) + To*cos(lamda4) - To*cos(lamda5)
```

$$F2\cos(phi1 + theta6) - F1\cos(theta6) + To\cos(lamda4) - To\cos(lamda5)$$

```
expre(2,1):=F2*sin(phi1 + theta6) - F1*sin(theta6) + To*sin(lamda4) + To*sin(lamda5)
```

$$F2\sin(phi1 + theta6) - F1\sin(theta6) + To\sin(lamda4) + To\sin(lamda5)$$

```
expre(3,1):=- F2*cos(phi1 + theta6) - F3*cos(phi2 + theta7) - Fs*cos(psiL)
```

$$-F2\cos(phi1 + theta6) - F3\cos(phi2 + theta7) - Fs\cos(psiL)$$

```
expre(4,1):=- F2*sin(phi1 + theta6) - F3*sin(phi2 + theta7) - Fs*sin(psiL)
```

$$-F2\sin(phi1 + theta6) - F3\sin(phi2 + theta7) - Fs\sin(psiL)$$

```
expre(5,1):=F3*cos(phi2 + theta7) - F4*cos(theta7) + To*cos(3*PI/2 + lamda2) + To*cos(PI/2 - lamda3)
```

$$F3\cos(phi2 + theta7) - F4\cos(theta7) + To\cos\left(\frac{3\pi}{2} + lamda2\right) + To\cos\left(\frac{\pi}{2} - lamda3\right)$$

```
expre(6,1):=To*sin(PI/2 - lamda3) + F3*sin(phi2 + theta7) - F4*sin(theta7) + To*sin((3*PI)/2 + lamda2)
```

$$To\sin\left(\frac{\pi}{2} - lamda3\right) + F3\sin(phi2 + theta7) - F4\sin(theta7) + To\sin\left(\frac{3\pi}{2} + lamda2\right)$$

```
M:= Dom::Matrix(): // Define matrix domain
```

```
b:=M(6,1):  // initialize b as matrix
```

```
for row from 1 to 6 do
   b[row,1]:=-1*expre(row,1)|{F1=0,F2=0,F3=0,F4=0,Fs=0}:
end_for
```

$$-To\sin\left(\frac{\pi}{2} - lamda3\right) - To\sin\left(\frac{3\pi}{2} + lamda2\right)$$

```
b   // display b below:
```

$$\begin{pmatrix} To\cos(lamda5) - To\cos(lamda4) \\ -To\sin(lamda4) - To\sin(lamda5) \\ 0 \\ 0 \\ -To\cos\left(\frac{3\pi}{2} + lamda2\right) - To\cos\left(\frac{\pi}{2} - lamda3\right) \\ -To\sin\left(\frac{\pi}{2} - lamda3\right) - To\sin\left(\frac{3\pi}{2} + lamda2\right) \end{pmatrix}$$

```
A:=M(6,5): // initialize A matrix

for row from 1 to 6 do
   A[row,1]:=(expre(row,1)+b[row,1])|{F1=1,F2=0,F3=0,F4=0,Fs=0}:
   A[row,2]:=(expre(row,1)+b[row,1])|{F1=0,F2=1,F3=0,F4=0,Fs=0}:
   A[row,3]:=(expre(row,1)+b[row,1])|{F1=0,F2=0,F3=1,F4=0,Fs=0}:
   A[row,4]:=(expre(row,1)+b[row,1])|{F1=0,F2=0,F3=0,F4=1,Fs=0}:
   A[row,5]:=(expre(row,1)+b[row,1])|{F1=0,F2=0,F3=0,F4=0,Fs=1}:
//   print(Unquoted,"a". expr2text(row)."1=".expr2text(A(row,1)));
//   print(Unquoted,"a". expr2text(row)."2=".expr2text(A(row,2)));
//   print(Unquoted,"a". expr2text(row)."3=".expr2text(A(row,3)));
//   print(Unquoted,"a". expr2text(row)."4=".expr2text(A(row,4)));
//   print(Unquoted,"a". expr2text(row)."5=".expr2text(A(row,5)));
end_for
 0
```

```
A   // display A below:
```

$$\begin{pmatrix} -\cos(\text{theta6}) & \cos(\text{phi1}+\text{theta6}) & 0 & 0 & 0 \\ -\sin(\text{theta6}) & \sin(\text{phi1}+\text{theta6}) & 0 & 0 & 0 \\ 0 & -\cos(\text{phi1}+\text{theta6}) & -\cos(\text{phi2}+\text{theta7}) & 0 & -\cos(\text{psiL}) \\ 0 & -\sin(\text{phi1}+\text{theta6}) & -\sin(\text{phi2}+\text{theta7}) & 0 & -\sin(\text{psiL}) \\ 0 & 0 & \cos(\text{phi2}+\text{theta7}) & -\cos(\text{theta7}) & 0 \\ 0 & 0 & \sin(\text{phi2}+\text{theta7}) & -\sin(\text{theta7}) & 0 \end{pmatrix}$$

3 - DEMONSTRATE THE SYNTAX FOR DELETING A ROW (IN THIS EXAMPLE 2ND ROW)

```
M::delRow(A, 2)    // show A without 2nd row
```

$$\begin{pmatrix} -\cos(\text{theta6}) & \cos(\text{phi1}+\text{theta6}) & 0 & 0 & 0 \\ 0 & -\cos(\text{phi1}+\text{theta6}) & -\cos(\text{phi2}+\text{theta7}) & 0 & -\cos(\text{psiL}) \\ 0 & -\sin(\text{phi1}+\text{theta6}) & -\sin(\text{phi2}+\text{theta7}) & 0 & -\sin(\text{psiL}) \\ 0 & 0 & \cos(\text{phi2}+\text{theta7}) & -\cos(\text{theta7}) & 0 \\ 0 & 0 & \sin(\text{phi2}+\text{theta7}) & -\sin(\text{theta7}) & 0 \end{pmatrix}$$

```
M::delRow(b,2)   // show b without 2nd row
```

$$\begin{pmatrix} \text{To}\cos(\text{lamda5}) - \text{To}\cos(\text{lamda4}) \\ 0 \\ 0 \\ -\text{To}\cos(\frac{3\pi}{2}+\text{lamda2}) - \text{To}\cos(\frac{\pi}{2}-\text{lamda3}) \\ -\text{To}\sin(\frac{\pi}{2}-\text{lamda3}) - \text{To}\sin(\frac{3\pi}{2}+\text{lamda2}) \end{pmatrix}$$

4 - GENERATE SOLUTION MATRICES BY DELETING ONE ROW AT A TIME

Generate vectors X1, X2, X3 etc which correspond to solution which is found by deleting the 1st, 2nd, 3rd etc equation. X = A_inverse * b

```
X1:=M::delRow(A, 1)^-1 * M::delRow(b, 1):
X2:=M::delRow(A, 2)^-1 * M::delRow(b, 2):
X3:=M::delRow(A, 3)^-1 * M::delRow(b, 3):
X4:=(M::delRow(A, 4)^-1) * M::delRow(b, 4):
X5:=(M::delRow(A, 5)^-1) * M::delRow(b, 5):
X6:=(M::delRow(A, 6)^-1) * M::delRow(b, 6):
```

5 - COMPARE RESULTS USING ARBITRARILY SELECTED VALUES FOR ANGLES AND To

Choose random set of values with which to examine these matrices

```
values:={phi1=0.05,phi2=0.1, lamda2=0.15,lamda3=0.2,
  lamda4=0.25,lamda5=0.3,theta6=0.35,theta7=0.4,psiL=0.45,
  To=1}
```

$$\{\text{lamda4} = 0.25,\ \text{lamda2} = 0.15,\ \text{lamda5} = 0.3,\ \text{theta6} = 0.35,\ \text{lamda3} = 0.2,\ \text{phi1} = 0.05,\ \text{phi2} = 0.1,\ \text{theta7} = 0.4,\ \text{psiL} = 0.45,\ \text{To} = 1\}$$

```
A|values   // full A6x5 with values plugged in
```

$$\begin{pmatrix} -0.9393727128 & 0.921060994 & 0 & 0 & 0 \\ -0.3428978075 & 0.3894183423 & 0 & 0 & 0 \\ 0 & -0.921060994 & -0.8775825619 & 0 & -0.9004471024 \\ 0 & -0.3894183423 & -0.4794255386 & 0 & -0.4349655341 \\ 0 & 0 & 0.8775825619 & -0.921060994 & 0 \\ 0 & 0 & 0.4794255386 & -0.3894183423 & 0 \end{pmatrix}$$

```
float(b|values)  // full b6x1 with values plugge din
```

$$\begin{pmatrix} -0.01357593259 \\ -0.5429241659 \\ 0 \\ 0 \\ -0.3481074633 \\ 0.008704500095 \end{pmatrix}$$

BELOW ARE SOLUTIONS (F1, F2, F3, F4, Fs) for these values, first with 1st row deleted, then with 2nd row etc

```
float(X1|values)
```

$$\begin{pmatrix} 3.216618796 \\ 1.43816381 \\ 1.43816381 \\ 1.748217495 \\ -2.872732958 \end{pmatrix}$$

```
float(X2|values)  // differs from x1 in F1 solution
```

$$\begin{pmatrix} 1.424581002 \\ 1.43816381 \\ 1.43816381 \\ 1.748217495 \\ -2.872732958 \end{pmatrix}$$

```
float(X3|values)
```

$$\begin{pmatrix} -9.899715461 \\ -10.11127229 \\ 1.43816381 \\ 1.748217495 \\ 7.467309886 \end{pmatrix}$$

```
float(X4|values) // differs from X3 in the Fs solution
```

$$\begin{pmatrix} -9.899715461 \\ -10.11127229 \\ 1.43816381 \\ 1.748217495 \\ 8.941103814 \end{pmatrix}$$

```
float(X5|values)
```

$$\begin{pmatrix} -9.899715461 \\ -10.11127229 \\ -10.11127229 \\ -12.47066749 \\ 20.19727166 \end{pmatrix}$$

```
float(X6|values) // differs from X5 in the F4 solution
```

$$\begin{pmatrix} -9.899715461 \\ -10.11127229 \\ -10.11127229 \\ -9.25603063 \\ 20.19727166 \end{pmatrix}$$